

# On Amplitude and Phase in Printed Characters

Bernt C. Skottun, Skottun Research, Piedmont, CA

The purpose of the project described here is to provide a graphic demonstration of the role of phases and amplitudes in determining the appearance of printed characters. The images of two printed characters are Fourier analyzed and the amplitude spectrum of one is combined with the phase spectrum of the other. In the resulting hybrid image, one can clearly recognize the letter whose phase spectrum is used, while no trace of the other letter is evident. This procedure is also carried out with letters of different fonts and sizes. Again, the resulting images are mainly determined by the phase spectra. However, with letters of different sizes, the amplitude spectra, as one might expect, influence the appearance.

The Fourier transform decomposes a signal into component waves of different frequencies. Each component is specified by its amplitude and phase. Two-dimensional images can be analyzed in the same way. In the case of naturalistic scenes, Oppenheim and Lim [1981], Piotrowski and Campbell [1982], and Shapley et al. [1990] have demonstrated that phase spectra are more important than amplitude spectra in determining the appearance of the image. Duan and Morris [1989] have indicated that this is also the case for printed text.

The project described here uses *Mathematica* to provide a graphic demonstration of the relative importance of amplitudes and phases in determining the appearance of printed characters. The approach is similar to the one used by Oppenheim and Lim [1981], and Piotrowski and Campbell [1982]. Hybrid images, containing the amplitudes of one character and the phases of another, are synthesized. From these images, the viewer can deduce the relative importance of amplitude and phase based on which character the hybrid image resembles.

The first step is to obtain bitmap images of some printed characters. This is easily done using the *Mathematica* notebook interface. We can generate a PostScript image with an input such as:

```
In[1]:= Show[Graphics[Text[
  FontForm["b", {"Times-Bold", 64}], {0, 0}]]]
```



---

Bernt C. Skottun holds a Ph.D. in Physiological Optics from the University of California, Berkeley. His company, Skottun Research, develops medical tools and devices.

Applying the **Convert to Bitmap** and **Convert to Input-Form** commands to the PostScript cell yields an expression of the form `Show[Graphics[Raster[large array]]]`. After evaluating this expression, the bitmap array can be extracted using:

```
In[2]:= b1 = %[[1, 1, 1]];
```

The array consists of zeros and ones and has dimensions

```
In[3]:= Dimensions[b1]
Out[3]:= {175, 282}
```

We take the central  $64 \times 64$  block, which contains the image of the character, and save it to a file:

```
In[4]:= b2 = Take[ Take[#, {109, 172}]& /@ N[b1, 1], {55, 118}];
```

(The function `N` here converts the entries from the default precision to the form 1. or 0.)

```
In[5]:= Dimensions[b2]
Out[5]:= {64, 64}
In[6]:= b2 >> "b64"
```


Files containing bitmap arrays for the letters **b** and **g** are included in the electronic supplement. These letters were selected because (in the Times font) they seem intuitively to be very different: **b** contains dominant vertical contours while **g** appears to be characterized by curves and horizontal contours. The Times font is a good representative of most printed text.

To create the hybrid images, read in the bitmaps for the two characters:

```
In[1]:= b = << "b64"; g = << "g64";
```

The following function displays bitmap images:

```
In[2]:= display[Letters_List] :=
  Show[GraphicsArray[
    ListDensityPlot[#, FrameTicks -> None,
      Mesh->False, DisplayFunction -> Identity]& /@
    Letters]]
In[3]:= display[{b, g}]
```




The next step is to take the Fourier transform of the characters and extract their amplitudes and phases:

```
In[4]:= {fb, fg} = Fourier /@ {b, g};
In[5]:= {ampb, ampg} = Abs /@ {fb, fg};
In[6]:= {phaseb, phaseg} = Arg /@ {fb, fg};
```

Two hybrid images are created by combining the phases of one character with the amplitudes of the other and then taking the inverse Fourier transform:

```
In[7]:= gb = ampg Exp[I phaseb] // InverseFourier // Chop;
In[8]:= bg = ampb Exp[I phaseg] // InverseFourier // Chop;
In[9]:= display[{gb, bg}]
```




The image on the left combines the phase spectrum of **b** and the amplitude spectrum of **g**. The image on the right combines the phases of **g** with the amplitudes of **b**. In both cases, one can clearly recognize the letter whose phase spectrum was used and one sees no trace of the other letter.

It has been suggested that such apparent dominance of phase spectra may simply reflect a high degree of similarity of the amplitude spectra [Juvells et al. 1991; Tadmor and Tolhurst 1993]. For the set of letters {b, d, g, l, n o, u}, the correlations between the amplitude spectra were calculated, revealing that the amplitude spectra are indeed very similar ( $r$  is typically larger than 0.97). The apparent domination of

phase over amplitude may therefore mainly reflect the fact that there is little information in the amplitude spectra to distinguish the letters.


To further demonstrate the relative importance of the phase spectrum, we combine it with a flat amplitude spectrum (setting all amplitudes to 1) and take the inverse Fourier transform:

```
In[10]:= bp = Exp[I phaseb] // InverseFourier // Chop;
In[11]:= gp = Exp[I phaseg] // InverseFourier // Chop;
In[12]:= display[{bp, gp}]
```



In these images, which contain phase information only, one can clearly see the outline of each letter. On the other hand, setting the phases to zero and taking the inverse transform of the amplitude spectrum gives images in which no letters are recognizable:

```
In[13]:= ba = ampb // InverseFourier // Chop;
In[14]:= ga = ampg // InverseFourier // Chop;
In[15]:= display[{ba, ga}]
```




These images suggest that the phase spectra are genuinely more important than amplitude spectra for the appearance of characters and that the result of the initial test is not simply a reflection of the similarity of the amplitude spectra.

Morris [1992] has suggested that the typeface of printed text can be identified from amplitude information. Typefaces differ in two ways, in style and size. Differences in typeface style are found mainly in the details of the characters (such as serifs and widths of parts of the letters). In the case of naturalistic images, it has been suggested that amplitude spectra contain information about texture [Morgan et al. 1991; Tadmor and Tolhurst 1993]. Texture involves fine details and may be related to the fine details that distinguish different

styles of typefaces. As for character size, one would expect characters of different sizes to differ in their amplitude spectra, with larger characters having their amplitudes shifted to lower frequencies. (This should apply at least as a first approximation; however, changing character size is not simply a matter of scaling, but also involves subtle changes in the proportions of the characters.) In both style and size, one might expect to find a contribution of amplitude to the appearance of the typeface.


The files `timesk` and `helvk` contain bitmaps of the letter `k` in two different font types, Times and Helvetica:

```
In[16]:= kt = << "timesk"; kh = << "helvk";
In[17]:= display[{kt, kh}]
```



Here are the hybrid images constructed from the phase and amplitude spectra of these characters:


```
In[18]:= {fkt, fkh} = Fourier /@ {kt, kh};
In[19]:= display[Chop[
  {InverseFourier[Abs[fkh] Exp[I Arg[fkt]]],
  InverseFourier[Abs[fkt] Exp[I Arg[fkh]]}] ]];
```




The image on the left is obtained with the phase spectrum of the Times character and the amplitude of the Helvetica character. It clearly shows the appearance of the Times font type. Conversely, the image on the right combines the phase spectrum of the Helvetica character with the amplitude spectrum of the Times character. Again, the appearance is determined by the phase spectrum.

Finally, we construct hybrid images from a character in two sizes:

```
In[20]:= bb = << "b18";
In[21]:= display[{b, bb}]
```



```
In[22]:= fbb = Fourier[bb];
In[23]:= display[Chop[
  {InverseFourier[Abs[fbb] Exp[I Arg[fbb]]],
  InverseFourier[Abs[fbb] Exp[I Arg[fbb]]}] ]];
```



The image on the left combines the phases of the large letter with the amplitudes of the small letter. One can clearly recognize the outline of the large character. However, there are obvious lighter areas within the strokes of the letters. This is reminiscent of the effect of a high-pass filter, which may be an approximation to the result we have obtained since the amplitude spectrum of the small letter differs from that of the large letter mainly in that the components have been shifted to higher frequencies. Conversely, one would expect the combination of the phases of the small letter with the amplitudes of the large letter to yield an effect similar to a low-pass filtering of the small letter. It is a little surprising, therefore, that the amplitude spectra of the letters above are very similar, with a correlation coefficient of  $r = 0.965$ . Although the image on the right resembles a low-pass filtering of the small `b`, the images above are quite different in appearance, demonstrating that small changes in amplitude spectra can produce marked changes in the image. In summary, therefore, we conclude that while the appearance of printed characters is dominated by the phase spectra, even to the extent that letters with flat amplitude spectra can be readily recognized, relatively small changes in amplitude spectra may markedly influence their appearance.

The experiments described here were motivated by a desire to understand dyslexia by studying the demands that reading places on the visual system. Based on the present results, one might expect that an individual with limited ability to discriminate spatial phase would have difficulty reading. There are additional reasons to expect that deficiencies in phase discrimination cause problems that are particularly pronounced when reading. Each cortical neuron only responds to stimuli within its receptive field, a limited part of the whole visual field. Therefore, the Fourier analysis performed

by these neurons is a local, or “patchwise,” analysis. Morgan et al. [1991] have shown that amplitudes are much more important when analysis is performed in patches of a larger image. For this reason, amplitudes may contribute significantly to the appearance of visual scenes that are much larger than the individual receptive fields. Individual printed characters, on the other hand, subtend visual angles comparable to the sizes of the receptive fields of single cortical neurons (under normal reading conditions). Therefore, phase information may be specifically important in discriminating individual printed characters.

Some perceptual tests have indicated that the ability of the human visual system to discriminate phases is relatively poor [Burr 1980]. It is possible, therefore, that the visual system is operating near its limit when reading. If that is so, a small reduction in the ability to discriminate phase information may result in confusion of printed characters, as is the case in dyslexia, while performance of other visual tasks may be close to normal.

## References

- Burr, D.C. 1980. Sensitivity to spatial phase. *Vision Research* 20:391–396.
- Duan, G., and R.A. Morris. 1989. The importance of phase in the spectra of digital type. *Electronic Publishing* 2:47–59.
- Juvells, I., S. Vallmitjana, A. Carnicer, and J. Campos. 1991. The role of amplitude and phase of the Fourier transform in the digital image processing. *American Journal of Physics* 59:744–748.
- Morgan, M.J., J. Ross, and A. Hayes. 1991. The relative importance of local phase and local amplitude in patchwise image reconstruction. *Biological Cybernetics* 65: 113–119.
- Morris, R.A. 1992. Classification of digital typefaces using spectral signatures. *Pattern Recognition* 25:869–876.
- Oppenheim, A.V., and J.S. Lim. 1981. The importance of phase in signals. *Proceedings of the IEEE* 69:529–541.
- Piotrowski, L.N., and F.W. Campbell. 1982. A demonstration of the visual importance and flexibility of spatial-frequency amplitude and phase. *Perception* 11:337–346.
- Shapley, R., T. Caelli, S. Grossberg, M. Morgan, and I. Rentschler. 1990. Computational theories of visual perception. In L. Spillman and J.S. Werner (eds.), *Visual perception: The Neurophysiological Foundations*, 417–447. New York, Academic Press.
- Tadmor, Y., and D.J. Tolhurst. 1993. Both the phase and the amplitude spectrum may determine the appearance of natural images. *Vision Research* 33:141–145.

Bernt C. Skottun  
 Skottun Research, 273 Mather Street,  
 Piedmont, CA 94611  
 bernt@valois.berkeley.edu



The electronic supplement contains the image files  
 b64, g64, b18, timesk, and helvk.